ABSTRACT

SAS provides many methods to perform a table lookup like Merge Statement, Proc SQL, Format. Starting with SAS 9.3, Even we can define our custom routine to perform table lookup since hashing is available in user-defined subroutines through the FCMP procedure. By embedding a hash object in PROC FCMP functions, simplicity in program can be enhanced. This paper introduces table lookup using HASH in PROC FCMP and demonstrate a practical application in using this technic to solve real world problems.

INTRODUCTION

Hashing is a search algorithm to index a table, thus to store and retrieve associated data by using index hash key. It has been introduced into SAS® since SAS 9. Being considered the fastest way to search large amount information, Hash’s main usage is to improve program performance when working with large dataset. While hashing is often used through DATA step hash object in SAS Program, its syntax is often considered complicated and hard to maintain.

The FCMP procedure enables you to create user-defined SAS functions using DATA step syntax that is stored in a data set, this function can be called from the DATA step just as you would any other SAS function. Comparing with modern programming language, such as Python, R, SAS function has very limitation in general programming activity.

However, starting with SAS 9.3, hashing is available in user-defined subroutines through the FCMP procedure. By embedding HASH in FCMP, we can extend the functionality of user-defined SAS functions and join the power of HASH and FCMP. This could streamline your existing programs, make your program more generic, and enhance simplicity. This paper is not intent to teach fundamental HASH or FCMP knowledge, but to introduce a new way to implement HASH and FCMP

TABLE LOOKUP – TRADITIONAL APPROACH VS HASHING FCMP

There are many ways to look up table: merge, SQL join, Format, Data hashing, and etc. Below is an example comparing using traditional approach and hashing FCMP approach in table lookup.

<table>
<thead>
<tr>
<th>state</th>
<th>capital</th>
<th>state</th>
</tr>
</thead>
<tbody>
<tr>
<td>North Carolina</td>
<td>Raleigh</td>
<td>Virginia</td>
</tr>
<tr>
<td>Virginia</td>
<td>Richmond</td>
<td>2</td>
</tr>
<tr>
<td>South Carolina</td>
<td>Columbia</td>
<td>3</td>
</tr>
<tr>
<td>North Carolina</td>
<td>Little Rock</td>
<td>4</td>
</tr>
<tr>
<td>Alabama</td>
<td>Montgomery</td>
<td>5</td>
</tr>
<tr>
<td>Louisiana</td>
<td>Baton Rouge</td>
<td>6</td>
</tr>
<tr>
<td>Tennessee</td>
<td>Nashville</td>
<td>7</td>
</tr>
</tbody>
</table>

Figure 1. Sates and Capital Sample Data

Traditional Approach

1. Data Step

/* Data Step approach*/

proc sort data = StatesCapital; by state; run;
proc sort data = States; by state; run;
data new;
  merge StatesCapital States;
  by state;
run;
Table lookup using HASH in FCMP, continued

2. PROC SQL

/* Proc SQL */
proc sql;
create table new as
select a.state, b.capital
from States as a, StatesCapital as b
where a.state = b.state;
quit;

3. Data Hash

/* Data Hash */
data new;
if 0 then set work.StatesCapital;
if _n_ = 1 then do;
   declare hash hh (dataset: "work.StatesCapital");
   hh.definedata("capital");
   hh.definekey("state");
   hh.definedone();
end;
set States;
rc =hh.find();
if rc = 0 then capital = capital;
run;

Hash FCMP

While many SAS users use Data Step or SQL to perform table lookup in their daily work. Advanced programmer may use hash in their program for efficiency purpose. However, using hash means more syntax, thus the program is much complex and hard to maintain. In this case, wrapping hash in SAS user-defined functions provide a more generic and simple way. And program could be much simplified.

/* Define the hash function */
proc fcmp outlib=work.functions.samples;
function hash_fcmp(state $) $25;
   declare hash hh (dataset: "work.StatesCapital");
   rc=hh.definedata("capital");
   rc=hh.definekey("state");
   rc=hh.definedone();
   rc=hh.find();
   if rc eq 0 then return(capital);
   else return('Not Found');
endsub;
quit;
options cmplib=work.functions;

/* Call the function */
* In Data Step;
data new;
   set states;
   capital = hash_fcmp(state);
run;

* In Proc SQL;
proc sql;
create table new as
select *,hash_fcmp(state) as capital
from states;
quit;

As we can see in above example, embedding HASH in FCMP made table lookup more elegant. By packaging HASH in FCMP, the complicated HASH syntax goes no way.

**FCMP HASH OBJECT METHOD AND SYNTAX**

<table>
<thead>
<tr>
<th>Method</th>
<th>Syntax</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>DECLARE</td>
<td>DECLARE hash_object-name</td>
<td>Create a new instance of hash object, create at parse time.</td>
</tr>
<tr>
<td>DEFINEKEY</td>
<td>rc = object.DEFINEKEY('key 1','key n')</td>
<td>Set up key variables for hash object</td>
</tr>
<tr>
<td>DEFINEDATA</td>
<td>rc = object.DEFINEDATA('dataset 1','dataset n')</td>
<td>Define data to be stored in hash object</td>
</tr>
<tr>
<td>DEFINEDONE</td>
<td>rc = object.DEFINEDONE()</td>
<td>Indicate the key and data specification is completed</td>
</tr>
<tr>
<td>DELETE</td>
<td>rc = object.DELETE()</td>
<td>Delete the hash object and free any resources allocated</td>
</tr>
<tr>
<td>FIND</td>
<td>rc = object.FIND('key1','keyn')</td>
<td>Search a hash object based on the values of defined key, if look up is successful, defined data variable are updated</td>
</tr>
<tr>
<td>CHECK</td>
<td>rc = object.CHECK()</td>
<td>Search a hash object based on the values of defined key, data will not be updated whether If look up is successful</td>
</tr>
<tr>
<td>NUM_ITEMS</td>
<td>rc = object.NUM_ITEMS()</td>
<td>Return the number of items in hash object</td>
</tr>
<tr>
<td>ADD</td>
<td>rc = object.ADD(key: value1, key: value n)</td>
<td>Add data with associated key to hash object</td>
</tr>
<tr>
<td>REMOVE</td>
<td>rc = object.REMOVE(key: value1, key: value n)</td>
<td>Remove data with associated key to hash object</td>
</tr>
</tbody>
</table>

Table 2. HASH in FCMP Method and Syntax

**PRACTICAL APPLICATION**

In a Proportional font, such as the font this article is set in, different letters occupy different amount of horizontal space. For instance, the letter “I” is much narrower than the letter “W”. Most books, magazines and other printed materials are set in proportional fonts. similarly, the graphical user interface of many programs uses a proportional font for titles, menus and other text. Examples of commonly-used proportional fonts are Times New Roman, Verdana, Arial, Georgia and Comic Sans. For my experience, there are no exist method in SAS to get a string widths directly. The method used in the following example demonstrate a way to calculate the display width of a proportional font string by joining the power of HASH and PROC FCMP.

**WHY SHOULD WE GET PRINT WIDTH OF A STRING**

The idea to calculate the display width of a string comes from two of my projects.

The one is using SAS to draw annotation in rectangular box in CRF which is in PDF format in the development of a aCRF automation tool, and we need to decide how long the rectangular box should be. The width of rectangular box varied from different string. Unfortunately, the string needs in proportional fonts and we can’t get the rectangular box width just by counting characters in string.

**DM = Demographics**

Figure 2. Determine the width of rectangular box

Another is to avoid unwanted wrap up in RTF output since SAS let Microsoft Word to determine the print width. We need to get exact column width for each column so that text would not wrap up. This is especially important in some report which contains many columns and we cannot simply imagine the column width.

Figure 3. Determine the width of RTF output
THE COMPONENTS OF CHARACTER WIDTH

A: the width to add to the current position before placing the character
B: the width of the character itself
C: the white space to the right of the character
The total width is determined by calculating the sum of A + B + C

Figure 4. Characters Width

SAS APPROACH TO GET WIDTH

The process to get a print character width is listed below.

1. Determine the formula: \( \text{String Width} = \text{Sum(Unique Character Width)} \)
2. Got unique character width and stored in table
3. Retrieve using HASH and sum up using FCMP

Get unique Character width

Completed tables regarding print character width is not available in the website, thus I made a method approximating character width by using SAS. And it worked well. The method is listed below.

1. Output characters in RTF format in specific style, and make sure one character fill with a line (Code could be found in the appendix)

2. Read RTF file back line by line, approximate character width using the following formula
   \[ \text{Width} = \frac{\text{Length of the Line}}{\text{Count of Character}} \]
3. Summarize The Hash Table by repeating above steps for different font style. Store the data in a dictionary table

Figure 4. The character width dictionary
MAKE THE HASH FUNCTION

* Customed SAS Function;

```
proc fcmp outlib=width.functions.GetWidth;
  function GetWidth(string $ ,_font_ $, _style_ $, _font_size_ $);
    length id $ 1;
    * Define Hash Table with Character Width Dataset;
    declare hash Calculate(dataset: "width.CharWidth");
    rc = Calculate.defineKey("char","_font_","_style_","_font_size_");
    rc = Calculate.defineData("_width_");
    rc = Calculate.defineDone();
    width_tot = 0;
    * Retrieve Data from Hash and Sum up;
    do i = 1 to lengthn(string);
      char = substr(string,i,1);
      rc   = Calculate.find();
      if rc eq 0 then width_tot = width_tot+_width_;
    end;
    return(width_tot);
endsub;
run;
```

SIMPLE USAGE

*------------------------ Hash in PROC FCMP Example Usage ------------------------;
* Using in Data Step;
```
data Width;
  set TEXT;
  CharWidth = GetWidth(Value,'Times','Normal','10');
run;
```

* Using in Proc SQL;
```
proc sql;
  create table Width as
    select GetWidth(Value,'Times','Normal','10') as CharWidth
    from TEXT;
quit;
```

* Using in Proc MEANS;
```
proc means data = TEXT;
  where GetWidth(Value,'Times','Normal','10') > 99;
  var Value;
run;
```

SIMPLE OUTPUT

<table>
<thead>
<tr>
<th>text</th>
<th>font</th>
<th>style</th>
<th>font_size</th>
<th>TextWidth</th>
</tr>
</thead>
<tbody>
<tr>
<td>MDX</td>
<td>Times</td>
<td>Normal</td>
<td>10</td>
<td>23.704258065</td>
</tr>
<tr>
<td>RSX Type S 2dr</td>
<td>Times</td>
<td>Normal</td>
<td>10</td>
<td>56.876593001</td>
</tr>
<tr>
<td>TSX 4dr</td>
<td>Times</td>
<td>Normal</td>
<td>10</td>
<td>34.967230228</td>
</tr>
<tr>
<td>TL 4dr</td>
<td>Times</td>
<td>Normal</td>
<td>10</td>
<td>28.180830925</td>
</tr>
<tr>
<td>3.5 RL 4dr</td>
<td>Times</td>
<td>Normal</td>
<td>10</td>
<td>43.313405145</td>
</tr>
<tr>
<td>3.5 RL w/Navigations 4dr</td>
<td>Times</td>
<td>Normal</td>
<td>10</td>
<td>101.3058582</td>
</tr>
<tr>
<td>NSX coupe 2dr manual S</td>
<td>Times</td>
<td>Normal</td>
<td>10</td>
<td>102.88078652</td>
</tr>
</tbody>
</table>
CONCLUSION

Table lookup by embedding Hash in PROC FCMP is not for efficiency, it’s to streamline your daily work, enhance program simplicity and extend the functionality of user-defined SAS functions. And repetitive coding can be minimized.

REFERENCES

- Sample 47224: Load a SAS data set into a Hash Object using PROC FCMP http://support.sas.com/kb/47/224.html
- PROC FCMP and DATA Step Component Objects http://documentation.sas.com/?docsetId=proc&docsetVersion=9.4&docsetTarget=n03uc8c8fkguxqn1i5apv1auqrz.htm&locale=en

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Qing Liu
Enterprise: Eli Lilly & Company
Address: 19F Tower 1 HKRI, Taikoo Hui, No. 288 Shi Men Yi Road
City, State ZIP: Shanghai, 200041
Work Phone: 13167017220
E-mail: MeetQingLiu@gmail.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.
Source Code Sample

* Dummy all characters in ASCII;

```sas
data null;
  length text $200;
  do i = 160, 32 to 127;
    text = repeat(byte(i), 300);
    output;
  end;
run;
```

* Define the output font style;

```sas
proc template;
  define style global.text;
    parent=styles.rtf;
    replace fonts /
      'fixedfont' = ("Times", 10pt, Bold Normal)
      'docfont' = ("Times", 10pt, Bold Normal)
    ;
  end;
run;
```

* Out RTF file and read this file back;

```sas
options orientation=portrait topmargin=1in bottommargin=1in leftmargin=1in rightmargin=1in nodate;
title '';
ods rtf file='test.rtf' style=global.text;
proc report data=null NOHEADER nowd;
  column text;
  define text/display;
run;
ods rtf close;
```

```sas
data width_1;
  infile "test.txt";
  input;
  length text $200;
  retain blank;
  if _n_ = 1 then blank=count(_infile_, '|');
  if ^missing(_infile_);
  text = strip(_ infile_);
  if _n_ ne 1;
run;
```

```sas
proc sql;
  create table width_2 as
    select distinct *, lengthn(text) as length
    from width_1
    group by substr(text, 1, 1)
    having lengthn(text)=max(lengthn(text));
quit;
```

```sas
data width_3;
  set width_2;
  width_tot = 6.3;
  if _n_ = 1 then do;
```

char = ' ';
width = (width_tot/blank)*72;
output;
end;
char = substr(text,1,1);
width = (width_tot/length)*72;
output;
keep char width;
run;
proc sort; by width; run;
data width;
set width_3;
length _font_ _style_ _font_size_ $20;
_font_ = 'Times';
_style_ = 'Bold Normal';
_font_size_ = '10';
_width_ = width;
keep char _font_ _style_ _font_size_ _width_; run;
data charwidth;
set width;
run;
* Customed SAS Function;
proc fcmp outlib=work.functions.GetWidth;
function GetWidth(string $ ,_font_ $, _style_ $, _font_size_ $);
length id $ 1;
* Retrieve Character Width Using HASH;
declare hash Calculate(dataset: "work.CharWidth");
rc = Calculate.defineKey("char","_font_","_style_","_font_size_");
rc = Calculate.defineData("_width_" );
rc = Calculate.defineDone();
width_tot = 0;
* Sum all width up for a string;
do i = 1 to lengthn(string);
    char = substr(string,i,1);
    rc = Calculate.find();
    if rc eq 0 then width_tot = width_tot+_width_; end;
return(width_tot);
endsub;
run;
options cmplib=work.functions;
data a;
   set sashelp.cars;
   b=GetWidth(strip(model), 'Times', 'Normal', '10');
run;